

```

<program>          ::= <statement_list>
<statement_list>   ::= <statement>
                    | <statement> <statement_list>
<statement>        ::= <load_stmt>
                    | <select_stmt>
                    | <filter_stmt>
                    | <sort_stmt>
                    | <join_stmt>
                    | <groupby_stmt>
                    | <sample_stmt>
                    | <dropna_stmt>
                    | <fillna_stmt>
                    | <mutate_stmt>
                    | <apply_stmt>
                    | <describe_stmt>
                    | <summary_stmt>
                    | <outliers_stmt>
                    | <quantile_stmt>
                    | <normalize_stmt>
                    | <binning_stmt>
                    | <rolling_stmt>
                    | <hypothesis_stmt>
                    | <boxplot_stmt>
                    | <heatmap_stmt>
                    | <pairplot_stmt>
                    | <timeseries_stmt>
                    | <pie_stmt>
                    | <save_stmt>
                    | <export_plot_stmt>
                    | <info_stmt>

```

```

<load_stmt>        ::= "load" STRING_LITERAL "as" IDENTIFIER
<select_stmt>      ::= "select" IDENTIFIER <column_list_curly> "as" IDENTIFIER
<filter_stmt>      ::= "filter" IDENTIFIER "[" <condition> "]" "as" IDENTIFIER
<sort_stmt>        ::= "sort" IDENTIFIER "by" ":" <column_list_sort> "as" IDENTIFIER
<join_stmt>        ::= "join" IDENTIFIER "with" ":" IDENTIFIER "on" ":" IDENTIFIER "as"
IDENTIFIER
<groupby_stmt>     ::= "groupby" IDENTIFIER "by" ":" <column_list_curly> "agg" ":"
<agg_func_list>    "as" IDENTIFIER
<sample_stmt>      ::= "sample" IDENTIFIER "n" ":" NUMERIC_LITERAL ["random"] "as"
IDENTIFIER
<dropna_stmt>      ::= "dropna" IDENTIFIER ["columns" ":" <column_list_curly>] "as"
IDENTIFIER
<fillna_stmt>      ::= "fillna" IDENTIFIER "value" ":" <literal> ["columns" ":"
<column_list_curly>] "as" IDENTIFIER
<mutate_stmt>      ::= "mutate" IDENTIFIER <mutation_list> "as" IDENTIFIER
<apply_stmt>       ::= "apply" IDENTIFIER "columns" ":" <column_list_curly> "function" ":"
STRING_LITERAL "as" IDENTIFIER

```

```

<describe_stmt> ::= "describe" IDENTIFIER ["columns" ":" <column_list_curly>]
<summary_stmt>  ::= "summary" IDENTIFIER
<outliers_stmt> ::= "outliers" IDENTIFIER "method" ":" IDENTIFIER "columns" ":"
<column_list_curly>
<quantile_stmt> ::= "quantile" IDENTIFIER "column" ":" IDENTIFIER "q" ":"
NUMERIC_LITERAL
<normalize_stmt> ::= "normalize" IDENTIFIER "columns" ":" <column_list_curly> "method"
":" IDENTIFIER "as" IDENTIFIER
<binning_stmt>   ::= "binning" IDENTIFIER "column" ":" IDENTIFIER "bins" ":"
NUMERIC_LITERAL "as" IDENTIFIER
<rolling_stmt>   ::= "rolling" IDENTIFIER "column" ":" IDENTIFIER "window" ":"
NUMERIC_LITERAL "function" ":" IDENTIFIER "as" IDENTIFIER
<hypothesis_stmt> ::= "hypothesis" IDENTIFIER "vs" ":" IDENTIFIER "columns" ":"
<column_list_curly> "test" ":" IDENTIFIER
<boxplot_stmt>   ::= "boxplot" IDENTIFIER "columns" ":" <column_list_curly>
<heatmap_stmt>   ::= "heatmap" IDENTIFIER "columns" ":" <column_list_curly>
<pairplot_stmt>  ::= "pairplot" IDENTIFIER "columns" ":" <column_list_curly>
<timeseries_stmt> ::= "timeseries" IDENTIFIER "x" ":" IDENTIFIER "y" ":" IDENTIFIER
<pie_stmt>       ::= "pie" IDENTIFIER "values" ":" IDENTIFIER "labels" ":" IDENTIFIER
<save_stmt>      ::= "save" IDENTIFIER "to" ":" STRING_LITERAL ["format" ":"
IDENTIFIER]
<export_plot_stmt> ::= "export_plot" "filename" ":" STRING_LITERAL ["width" ":"
NUMERIC_LITERAL] ["height" ":" NUMERIC_LITERAL]
<info_stmt>      ::= "info" IDENTIFIER

<condition>      ::= IDENTIFIER <operator> <literal>
                  | IDENTIFIER <operator> IDENTIFIER
<operator>       ::= "==" | "!=" | "<" | ">" | "<=" | ">="
<column_list_sort> ::= <sort_column_spec>
                  | <sort_column_spec> "," <column_list_sort>
<sort_column_spec> ::= IDENTIFIER ["desc"]
<column_list_curly> ::= "{" <column_names> "}"
<column_names>    ::= IDENTIFIER
                  | IDENTIFIER "," <column_names>
<agg_func_list>   ::= "{" <agg_functions> "}"
<agg_functions>   ::= <agg_function>
                  | <agg_function> "," <agg_functions>
<agg_function>    ::= IDENTIFIER ":" IDENTIFIER
<mutation_list>   ::= "{" <mutations> "}"
<mutations>       ::= <mutation>
                  | <mutation> "," <mutations>
<mutation>        ::= IDENTIFIER ":" STRING_LITERAL
<literal>         ::= STRING_LITERAL
                  | NUMERIC_LITERAL

```